
********************************************************************************
clear all
/*
cap log close
log using LOG/Import_Databases, replace smcl
*/
********************************************************************************

* Crear bases temporales:
forvalues i = 1995/2007 {
tempfile ENIA`i'_1
tempfile ENIA`i'_2
if `i' >= 2000 { // En los años sólo anteriores existen 2 partes.
tempfile ENIA`i'_3 
}
}

* Cargar bases de Excel:
forvalues k = 1995/2007 {

* Cargar bases dependiendo del nombre original del archivo:
if `k'< 1998 {
import excel SOURCE\BASE_FUSION_`k'_SE.xlsx, firstrow sheet("Fusion Parte 1 `k'") clear
save `ENIA`k'_1'
import excel SOURCE\BASE_FUSION_`k'_SE.xlsx, firstrow sheet("Fusion Parte 2 `k'") clear
save `ENIA`k'_2'
}

if `k'>=1998 & `k'<=2000 {
import excel SOURCE\BASE_FUSION_`k'_SE_ST.xlsx, firstrow sheet("Fusion Parte 1 `k'") clear
save `ENIA`k'_1'
import excel SOURCE\BASE_FUSION_`k'_SE_ST.xlsx, firstrow sheet("Fusion Parte 2 `k'") clear
save `ENIA`k'_2'
}

* Cargar bases para años más recientes que tienen 3 partes:
if `k'==2000 {
import excel SOURCE\BASE_FUSION_`k'_SE_ST.xlsx, firstrow sheet("Fusion Parte 3 `k'") clear
save `ENIA`k'_3'
}

if `k'>=2001 {
import excel SOURCE\Enia_`k'.xlsx, firstrow sheet("Fusion Parte 1 `k'") clear
save `ENIA`k'_1'
import excel SOURCE\Enia_`k'.xlsx, firstrow sheet("Fusion Parte 2 `k'") clear
save `ENIA`k'_2'
import excel SOURCE\Enia_`k'.xlsx, firstrow sheet("Fusion Parte 3 `k'") clear
save `ENIA`k'_3'
}

* Merge de las bases generadas para cada año
use `ENIA`k'_1'
merge 1:1 NUI Dvn using `ENIA`k'_2'

if `k' >= 2000 {
drop _merge
merge 1:1 NUI Dvn using `ENIA`k'_3'
}

drop _merge
gen YEAR = `k'
order YEAR, first
save DTA/ENIA_`k', replace
}

* Reemplazar algunos strings para poder hacer merge:
qui forvalues k = 1995/2007 {
use DTA/ENIA_`k', clear
qui destring, replace ignore(" ") // Variables reconocidas como string por tener espacios entre el NРy el signo.
tostring UCACAN UPECAN UBENCAN UPARCAN ULICCAN UCANCAN UNATCAN ULENCAN UGRCAN, replace // Strings ingresados como numero en los cohortes donde no habdatos.
if `k' == 2000 { 
tostring DHORA HHORA, replace 
}
save DTA/ENIA_`k', replace
}

* Unir todas las bases:
use DTA/ENIA_1995, clear
forvalues k = 1996/2007 {
di `k'
append using DTA/ENIA_`k'
}

save DTA/Datos_ENIA_95_07_Raw, replace

*cap log close
